NAG Library Routine Document

F07MWF (ZHETRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07MWF (ZHETRI) computes the inverse of a complex Hermitian indefinite matrix A, where A has been factorized by F07MRF (ZHETRF).

2 Specification

```
SUBROUTINE F07MWF (UPLO, N, A, LDA, IPIV, WORK, INFO)

INTEGER                 N, LDA, IPIV(*), INFO
COMPLEX (KIND=nag_wp) A(LDA,*), WORK(N)
CHARACTER(1)            UPLO
```

The routine may be called by its LAPACK name ***zhetri***.

3 Description

F07MWF (ZHETRI) is used to compute the inverse of a complex Hermitian indefinite matrix A, the routine must be preceded by a call to F07MRF (ZHETRF), which computes the Bunch–Kaufman factorization of A.

If UPLO = 'U', $A = PUDU^{\mathrm{H}}P^{\mathrm{T}}$ and A^{-1} is computed by solving $U^{\mathrm{H}}P^{\mathrm{T}}XPU = D^{-1}$ for X.

If UPLO = 'L', $A = PLDL^{\mathrm{H}}P^{\mathrm{T}}$ and A^{-1} is computed by solving $L^{\mathrm{H}}P^{\mathrm{T}}XPL = D^{-1}$ for X.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

1: UPLO – CHARACTER(1) *Input*

On entry: specifies how A has been factorized.

UPLO = 'U'
$A = PUDU^{\mathrm{H}}P^{\mathrm{T}}$, where U is upper triangular.

UPLO = 'L'
$A = PLDL^{\mathrm{H}}P^{\mathrm{T}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: A(LDA,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: details of the factorization of A, as returned by F07MRF (ZHETRF).

On exit: the factorization is overwritten by the n by n Hermitian matrix A^{-1}.

If UPLO = 'U', the upper triangle of A^{-1} is stored in the upper triangular part of the array.

If UPLO = 'L', the lower triangle of A^{-1} is stored in the lower triangular part of the array.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07MWF (ZHETRI) is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

5: IPIV($*$) – INTEGER array *Input*

Note: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

On entry: details of the interchanges and the block structure of D, as returned by F07MRF (ZHETRF).

6: WORK(N) – COMPLEX (KIND=nag_wp) array *Workspace*

7: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, $d(i,i)$ is exactly zero; D is singular and the inverse of A cannot be computed.

7 Accuracy

The computed inverse X satisfies a bound of the form

if UPLO = 'U', $\left|DU^{\mathrm{H}}P^{\mathrm{T}}XPU - I\right| \leq c(n)\epsilon\left(|D|\left|U^{\mathrm{H}}\right|P^{\mathrm{T}}|X|P|U| + |D|\left|D^{-1}\right|\right)$;

if UPLO = 'L', $\left|DL^{\mathrm{H}}P^{\mathrm{T}}XPL - I\right| \leq c(n)\epsilon\left(|D|\left|L^{\mathrm{H}}\right|P^{\mathrm{T}}|X|P|L| + |D|\left|D^{-1}\right|\right)$,

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

8 Further Comments

The total number of real floating point operations is approximately $\frac{8}{3}n^3$.

The real analogue of this routine is F07MJF (DSYTRI).

9 Example

This example computes the inverse of the matrix A, where

$$A = \begin{pmatrix} -1.36 + 0.00i & 1.58 + 0.90i & 2.21 - 0.21i & 3.91 + 1.50i \\ 1.58 - 0.90i & -8.87 + 0.00i & -1.84 - 0.03i & -1.78 + 1.18i \\ 2.21 + 0.21i & -1.84 + 0.03i & -4.63 + 0.00i & 0.11 + 0.11i \\ 3.91 - 1.50i & -1.78 - 1.18i & 0.11 - 0.11i & -1.84 + 0.00i \end{pmatrix}.$$

Here A is Hermitian indefinite and must first be factorized by F07MRF (ZHETRF).

9.1 Program Text

```
    Program f07mwfe

!     F07MWF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zhetrf, zhetri
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, lda, lwork, n
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,:), work(:)
      Integer, Allocatable             :: ipiv(:)
      Character (1)                    :: clabs(1), rlabs(1)
!     .. Executable Statements ..
      Write (nout,*) 'F07MWF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      lwork = 64*n
      Allocate (a(lda,n),work(lwork),ipiv(n))

!     Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)(a(i,1:i),i=1,n)
      End If

!     Factorize A
!     The NAG name equivalent of zhetrf is f07mrf
      Call zhetrf(uplo,n,a,lda,ipiv,work,lwork,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Compute inverse of A
!       The NAG name equivalent of zhetri is f07mwf
        Call zhetri(uplo,n,a,lda,ipiv,work,info)

!       Print inverse

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04dbf(uplo,'Nonunit',n,n,a,lda,'Bracketed','F7.4','Inverse', &
          'Integer',rlabs,'Integer',clabs,80,0,ifail)
```

```
      Else
        Write (nout,*) 'The factor D is singular'
      End If

    End Program f07mwfe
```

9.2 Program Data

```
F07MWF Example Program Data
  4                                                          :Value of N
  'L'                                                        :Value of UPLO
 (-1.36, 0.00)
 ( 1.58,-0.90) (-8.87, 0.00)
 ( 2.21, 0.21) (-1.84, 0.03) (-4.63, 0.00)
 ( 3.91,-1.50) (-1.78,-1.18) ( 0.11,-0.11) (-1.84, 0.00)  :End of matrix A
```

9.3 Program Results

```
 F07MWF Example Program Results

 Inverse
                     1                 2                 3                 4
 1  ( 0.0826, 0.0000)
 2  (-0.0335, 0.0440) (-0.1408, 0.0000)
 3  ( 0.0603,-0.0105) ( 0.0422,-0.0222) (-0.2007, 0.0000)
 4  ( 0.2391,-0.0926) ( 0.0304, 0.0203) ( 0.0982,-0.0635) ( 0.0073,-0.0000)
```